Reports of the Department of Geodetic Science and Surveying

Report No. 334

A FORTRAN PROGRAM FOR THE COMPUTATION OF GRAVIMETRIC QUANTITIES FROM HIGH DEGREE SPHERICAL HARMONIC EXPANSIONS

by

Richard H. Rapp

The Ohio State University
Department of Geodetic Science and Surveying
1958 Neil Avenue
Columbus, Ohio 43210

September, 1982

Abstract

The computer program described in this report calculates the height anomaly, gravity anomaly, gravity disturbance, and the two components of the deflection of the vertical using fully normalized potential coefficients of a spherical harmonic expansion. The program is designed to calculate these quantities on a point to point basis although certain calculations are not repeated if the latitude of the point does not change. The point input consists of the geodetic latitude, longitude, and height above the reference ellipsoid. Expansions up to degree 180 have been tested with the program.

The report first describes the theory to be implemented. The program is described with a set of results for five sample points computed with three different potential coefficient fields to degree 180. The computer time for a single point is 0.45 seconds per point on the Amdahl 470 V/8.

Foreword

This report was prepared by Richard H. Rapp, Professor, Department of Geodetic Science and Surveying, The Ohio State University, under Air Force Contract No. F19628-82-K-0022, The Ohio State University Research Foundation Project No. 714274. The contract covering this research is administered by the Air Force Geophysics Laboratory, Hanscom Air Force Base, Massachusetts, with Dr. Christopher Jekeli, Scientific Program Officer.

The reproduction and distribution of this report was carried out through funds supplied by the Department of Geodetic Science and Surveying. This report was also distributed by the Air Force Geophysics Laboratory as document AFGL-TR-82-0272, Scientific Report No. 1, under Contract No. F19628-82-K-0022.

Introduction

In the past few years the description of the earth's gravity potential in terms of spherical harmonic coefficients has been extended to degree 180 and in special cases to higher degrees (Rapp, 1978, 1981), and Lerch et al. (1981). These high degree expansions can be used to evaluate quantities such as geoid undulations, height anomalies, gravity anomalies, gravity disturbances, deflections of the vertical, etc. To do this efficient computer programs are needed. The purpose of this report is to describe one Fortran computer program that can be used for these calculations.

Theory--Basic Equations

The gravitational potential, V , in spherical harmonics can be written as:

$$V = \frac{kM}{r} [1 + \sum_{n=2}^{\infty} (\frac{a}{r})^n \sum_{m=0}^{n} (\bar{C}_{nm} \cos m\lambda + \bar{S}_{nm} \sin m\lambda) \bar{P}_{nm} (\sin\psi)] \quad (1)$$

where: kM is the geocentric gravitational constant;

r is the geocentric radius;

ψ is the geocentric latitude;

λ is the "geocentric" longitude;

$\bar{C}_{nm}, \bar{S}_{nm}$ are the fully normalized potential coefficients;

a is the scaling factor associated with the coefficients.

The disturbing potential, T , is the difference between the actual potential (V) at a point and the "normal" potential at the corresponding point. For our purpose the normal potential will be that associated with an equipotential reference ellipsoid of defined parameters. We have:

$$T(r,\psi,\lambda) = V(r,\psi,\lambda) - U(r,\psi,\lambda) \quad (2)$$

The potential associated with U can be described by an even degree zonal harmonic expansion. We can write:

$$T(r,\psi,\lambda) = \frac{(kM - kM_E)}{r} + \frac{kM}{r} \sum_{n=2}^{\infty} \left(\frac{a}{r}\right)^n \sum_{m=0}^{n} (\bar{C}^*_{nm} \cos m\lambda + \bar{S}_{nm} \sin m\lambda)\, \bar{P}_{nm}(\sin\psi) \quad (3)$$

where kM_E is the mass of the reference ellipsoid and $\bar{C}^*_{nm}$ are the differences between the actual coefficients and those implied by the reference equipotential ellipsoid. We have:

$$\begin{aligned} \bar{C}^*_{2,0} &= \bar{C}_{2,0} - \bar{C}_{2,0}(\text{ref}) \\ \bar{C}^*_{4,0} &= \bar{C}_{4,0} - \bar{C}_{4,0}(\text{ref}) \\ \bar{C}^*_{6,0} &= \bar{C}_{6,0} - \bar{C}_{6,0}(\text{ref}) \end{aligned} \quad (4)$$

In most cases we assume kM is equal to kM_E so that (3) becomes:

$$T(r,\psi,\lambda) = \frac{kM}{r} \sum_{n=2}^{\infty} \left(\frac{a}{r}\right)^n \sum_{m=0}^{n} (\bar{C}^*_{nm} \cos m\lambda + \bar{S}_{nm} \sin m\lambda)\, \bar{P}_{nm}(\sin\psi) \quad (5)$$

In classical gravimetric geodesy we discuss geoid undulations, N, and geop-spherop separations, N_r. If W_0 is the potential of the geoid and U_0 is the potential on the surface of the reference ellipsoid the geop-spherop separation is (Heiskanen and Moritz, 1967, Section 2-19):

$$N(r,\psi,\lambda) = \frac{Tr-(W_0 - U_0)}{\gamma\,(r,\psi)} \quad (6)$$

where γ is normal gravity. In most cases we take $W_0 = U_0$ so that (5) becomes:

$$N(r,\psi,\lambda) = \frac{T(r,\psi,\lambda)}{\gamma(r,\psi)} \quad (7)$$

The non-classical procedure uses the concept of the disturbance potential at some surface points and introduces the term height anomaly; ζ.

Let $W(r,\psi,\lambda)$ define the gravity potential and $U(r,\psi,\lambda)$ the normal gravity potential at the same point. Then:

$$T(r,\psi,\lambda) = W(r,\psi,\lambda) - U(r,\psi,\lambda) \tag{8}$$

We can introduce the geopotential number, C_P, with respect to a reference potential, W_0, such that

$$C_P = W(r,\psi,\lambda) - W_0 \tag{9}$$

The normal height of P, H^*, can be computed from C_P (Heiskanen and Moritz, 1967, section 4.5). Letting h be the geometric height of P above the reference ellipsoid the height anomaly is:

$$\zeta = h - H^* \tag{10}$$

In terms of the disturbing potential we can write:

$$\zeta = \frac{T(r,\psi,\lambda)}{\gamma(r,\psi)} \tag{11}$$

This equation is the same as (7) but there will be a conceptual (but small) difference when comparing normal heights, height anomalies, geoid undulations (N) and orthometric heights H. Specifically we have (ibid. section 8-12):

$$h = H + N = H^* + \xi \tag{12}$$

For our purposes we consider the disturbing potential to be given by equation (5) with the calculation of the height anomaly by (11). For the calculation of geoid undulations we would also use equation (11) but with the evaluation of T on the geoid by the appropriate choice of r. Although the convergence of the infinite series for T is a formal concern the calculations of Jekeli (1981) with high degree finite series show that there is no practical concern.

The gravity anomaly is a vector that can be expressed in the classical and non-classical forms. In either case the general relationship is the same between the disturbing potential and the anomaly although there is a conceptual difference. For the anomaly component in the vertical direction (h) we have (Heiskanen and Moritz, p. 967, p. 84, and 298):

$$\Delta g_h(r,\psi,\lambda) = -\frac{\partial T}{\partial h} + \frac{1}{\gamma}\frac{\partial \gamma}{\partial h} T(r,\psi,\lambda) \tag{13}$$

For the classical anomaly at the geoid, T is evaluated there, while for the surface anomaly T is evaluated at the surface point. We can obtain the radial component of the anomaly by writing (13) in the form:

$$\Delta g_r(r,\psi,\lambda) = \frac{-\partial T}{\partial r} + \frac{1}{\gamma}\frac{\partial \gamma}{\partial r} T(r,\psi,\lambda) \tag{14}$$

With a spherical approximation we have:

$$\frac{1}{\gamma}\frac{\partial \gamma}{\partial r} = -\frac{2}{r} \tag{15}$$

so that (14) becomes

$$\Delta g_r(r,\psi,\lambda) = -\frac{\partial T}{\partial r} - \frac{2}{r} T(r,\psi,\lambda) \tag{16}$$

If we now take equation (5) for T we have:

$$\Delta g_r(r,\psi,\lambda) = \frac{kM}{r^2}\sum_{n=2}^{\infty}(n-1)\left(\frac{a}{r}\right)^n \sum_{m=0}^{n}(\bar{C}_{nm}\cos m\lambda + \bar{S}_{nm}\sin m\lambda)\,\bar{P}_{nm}(\sin\psi) \tag{17}$$

The deflection of the vertical represents the angular difference between the normals to the actual gravity equipotential surface and the normal equipotential surface. For a deflection in an arbitrary direction (s) we can write (Pick et al., 1973, p. 257):

$$\theta = -\frac{1}{g}\frac{\partial T}{\partial s} \tag{18}$$

where s lies in the plane tangent to the normal equipotential surface. Normally the total deflections is expressed in a meridian (ξ) component and a prime vertical (η) component. In the meridian we have, with sufficient accuracy $ds = r d\psi$, and in the prime vertical, $ds = r \cos\psi d\lambda$. Thus the deflections of the vertical are:

$$\xi = -\frac{1}{gr}\frac{\partial T}{\partial \psi} , \quad \eta = -\frac{1}{g r \cos\psi} \frac{\partial T}{\partial \lambda} \tag{19}$$

As pointed out in Pick et als (1973, p. 307) the derivatives $\frac{\partial T}{\partial \psi}$ and $\frac{\partial T}{\partial \lambda}$ "are the derivatives of the disturbing potential with respect to the appropriate direction assuming that H and λ, and H and ϕ, respectively, are constant". (H corresponds to height and ϕ latitude.) Thus it is possible to use (5) for T to obtain the deflections. We then have, letting $g_P = \gamma(r,\psi)$:

$$\xi = -\frac{kM}{\gamma r^2} \sum_{n=2}^{\infty} \left(\frac{a}{r}\right)^n (\bar{C}^*_{nm} \cos m\lambda + \bar{S}_{nm} \sin m\lambda) \frac{d\,\bar{P}_{nm}(\sin\psi)}{d\psi} \tag{20}$$

$$\eta = -\frac{kM}{\gamma r^2 \cos\psi} \sum_{n=2}^{\infty} \left(\frac{a}{r}\right)^n \sum_{m=0}^{n} m(\bar{C}^*_{nm} (-\sin m\lambda) + \bar{S}_{nm} \cos m\lambda)\, \bar{P}_{nm}(\sin\psi) \tag{21}$$

To obtain the deflections in seconds multiply the above equations by the radian conversion factor.

The gravity disturbance vector is defined as the difference between gravity at a point and normal gravity at the same point. We have (Heiskanen and Moritz, 1967, p. 84):

$$\vec{\delta} = \vec{g}_P - \vec{\gamma}_P = \text{grad } T \tag{22}$$

The radial component of the gravity disturbance can be defined as (ibid, p. 85)

$$\delta_r = -\frac{\partial T}{\partial r} \tag{23}$$

In some cases (ibid, p. 233) the minus sign is not used but we retain (23) as our defining equation. Noting that (23) appears in (16) we can avoid a direct evaluation of δr by computing it from (16) after Δg and T (or $N(\zeta)$) have been computed. We have:

$$\delta r = \Delta g_r + \frac{2}{r} T \tag{24}$$

The other two components of the gravity disturbance vector are defined as (ibid, p. 285)

$$\delta_\psi = \frac{1}{r} \frac{\partial T}{\partial \psi}, \quad \delta_\lambda = \frac{1}{r \cos\psi} \frac{\partial T}{\partial \lambda} \tag{25}$$

Comparing these quantities to (19) we see

$$\xi = -\frac{1}{\gamma} \delta_\psi , \quad \eta = -\frac{1}{\gamma} \delta_\lambda \tag{26}$$

where we have let $g = \gamma$. Thus once ξ and η are computed it is a simple matter to calculate the two disturbance components δ_ψ , δ_λ .

In summary we are given a set of fully normalized potential coefficients. From these coefficients we remove the values implied by an equipotential reference ellipsoid. This leaves us with the expression for the disturbing potential T (equation 5). We then can compute height anomalies from equation (11), gravity anomalies from equation (17), the deflections of the vertical from (20) and (21), and the radial gravity disturbance from equation (24).

Theory--Auxilary Relationships

To implement the equations discussed in the previous section a number of additional quantities are needed. These are now discussed.

The Reference Potential Coefficients

Given four parameters defining an equipotential reference ellipsoid all the even degree zonal harmonics are explicitly defined. For our program it is sufficient to use only the zonal terms to degree six as taken from Cook (1959). We have given:

a = equatorial radius
kM = geocentric gravitational constant
ω = angular velocity
f = flattening.

Then compute m :

$$m = \frac{\omega^2 a^3 (1-f)}{kM} \tag{27}$$

The zonal coefficients in the J_n form are:

$$J_2 = \frac{2}{3}[f(1-\tfrac{1}{2}f) - \tfrac{1}{2}m(1 - \frac{2}{7}f + \frac{11}{49} f^2)] \tag{28}$$

$$J_4 = - \frac{4}{35} f(1-\tfrac{1}{2}f) (7f(1-\tfrac{1}{2}f) - 5m(1 - \frac{2}{7}f)) \tag{29}$$

$$J_6 = \frac{4}{21} f^2 (6f - 5m) \tag{30}$$

These coefficients are related to the fully normalized $\bar{C}$ coefficients through the following:

$$\bar{C}_{n0} = - \frac{J_n}{\sqrt{2n+1}} \tag{31}$$

<u>Calculation of ψ , and r</u>

Generally the latitude point will be specified as a geodetic latitude. Formally this latitude should be with respect to an ellipsoid whose center is at the center of mass of the earth. The geodetic latitude must be converted to a geocentric latitude and the geocentric radius must be computed. Given ϕ, λ, and h the rectangular coordinates of the point are (Rapp, 1981 equation 60).

$$\begin{aligned} X &= (N+h) \cos\phi \cos\lambda \\ Y &= (N+h) \cos\phi \sin\lambda \\ Z &= (N(1-e^2)+h) \sin\phi \end{aligned} \tag{32}$$

where N is the prime vertical radius of curvature:

$$N = \frac{a}{(1 - e^2 \sin^2\phi)^{\frac{1}{2}}} \tag{33}$$

The geocentric radius is then:

$$r = (X^2 + Y^2 + Z^2)^{\frac{1}{2}} \tag{34}$$

The geocentric latitude is then

$$\psi = \tan^{-1} \frac{Z}{\sqrt{X^2 + Y^2}} \tag{35}$$

Calculation of γ

Normal gravity is needed in the evaluation of the height anomaly and deflections of the vertical. A high degree of accuracy is not needed for this calculation as the number of digits in the final quantities is usually only two to four. In our case we choose to evaluate normal gravity for the point on the ellipsoid and then modify this value in a linear fashion for the height of the point above the ellipsoid. The normal gravity on the ellipsoid is:

$$\gamma = \gamma_E \frac{1 + k' \sin^2\phi}{\sqrt{1 - e^2 \sin^2\phi}} \tag{36}$$

The value of γ at the height h above the ellipsoid is:

$$\gamma_h = \gamma - 0.3086 \times 10^{-5}\, h \tag{37}$$

where γ is in meters/s^2 and h is in meters.

Calculation of $\bar{P}_{nm}$ and Its Derivative

The generation of the fully normalized associated Legendre functions and its first derivative is critical to any calculation involving spherical harmonic expansions. In choosing an algorithm one must consider the speed and the stability and accuracy of the procedure. In the past few years a number of different equation sets have been described in the literature.

For this program we have chosen subroutine LEGFDN that is described by Colombo (1981, p. 131). Colombo has carried out a number of tests to investigate the stability of the equations.

The subroutine is written such that the needed functions for a given order m and all degrees to the highest maximum degree are computed in one call to the subroutine. The subroutine is repeatedly called for $0 \leq m \leq N$ where N is the maximum degree being used in the expansion.

For discussion purposes visualize the associated Legendre functions in a lower triangular matrix where the rows correspond to degree n and the columns correspond to order m.

For a given m, the subroutine first calculates for $0 \leq n \leq m$ the diagonal elements corresponding to the diagonal passing through the $n = m$ location. We have:

$$
\begin{aligned}
&\bar{P}_{nm}(\cos\theta) = \sqrt{\frac{2n+1}{2n}}\ \sin\theta\ \bar{P}_{n-1,n-1}(\cos\theta) \\
&\bar{P}_{00}(\cos\theta) = 1.0 \qquad\qquad (38) \\
&\bar{P}_{11}(\cos\theta) = \sqrt{3}\ \sin\theta
\end{aligned}
$$

Then the following element is computed:

$$\bar{P}_{n+1,n}(\cos\theta) = \sqrt{2n+3}\ \cos\theta\ \bar{P}_{n,n}(\cos\theta) \qquad (39)$$

with $n = m$. Then the following recursive relationship is used to calculate the remaining values of $\bar{P}$ for $m + 2 \leq n \leq N$.

$$\bar{P}_{nm}(\cos\theta) = \sqrt{\frac{(2n-1)(2n+1)}{(n-m)(n+m)}} \cos\theta \, \bar{P}_{n-1,m}(\cos\theta)$$

$$- \sqrt{\frac{(2n+1)(n+m-1)(n-m-1)}{(2n-3)(n+m)(n-m)}} \, \bar{P}_{n-2,m}(\cos\theta)$$

$$n \geq 2 \; , \; (n-2) \geq m \geq 0 \tag{40}$$

Note that θ is the polar angle given by

$$\theta = 90° - \psi \tag{41}$$

Singh (1982) has pointed out ways to improve the calculation of the associated Legnedre functions by applying a scaling factor, such as 1×10^{72} in the recursive procedure. This procedure was tested in actual calculations of ξ , Δg etc. No difference in numberical values was seen when the scale factor was and was not applied. Consequently we did not implement the scaling operation. Doing so might avoid some underflow messages, but would not changes results when using expansions to degree 180.

For the ξ component calculations we need the derivative of $\bar{P}$. Colombo (1981) implemented the following procedure:

$$\frac{d\bar{P}_{nm}(\cos\theta)}{d\theta} = \left[\frac{2n+1}{2n}\right]^{\frac{1}{2}} \left(\sin\theta \frac{d\bar{P}_{n-1,n-1}}{d\theta} + \cos\theta \, \bar{P}_{n-1,n-1}(\cos\theta)\right) \tag{42}$$

After these values are computed for a given m up to a given N , then we have:

$$\frac{d\bar{P}_{nm}}{d\theta} = (\sin\theta)^{-1}\Big(n \, \bar{P}_{nm}(\cos\theta)\cos\theta$$

$$- \left[\frac{(n^2-m^2)(2n+1)}{(2n-1)}\right]^{-\frac{1}{2}} P_{n-1,m}(\cos\theta)\Big) \tag{43}$$

The starting value is

$$\frac{d\,\bar{P}_{00}}{d\theta} = 0 \tag{44}$$

Due to the occurrence of $(\sin\theta)^{-1}$ this subroutine can not calculate the derivatives at the poles.

Since we want the derivative of P with respect to ψ we note that:

$$\frac{d\bar{P}}{d\psi} = -\frac{d\bar{P}}{d\theta} \tag{45}$$

The calculation of $\sin m\lambda$ and $\cos m\lambda$

The generation of $\sin m\lambda$ and $\cos m\lambda$ is done through the following recursion relationships:

$$\begin{aligned} \sin m\lambda &= 2\cos\lambda \sin(m-1)\lambda - \sin(m-2)\lambda \\ \cos m\lambda &= 2\cos\lambda \cos(m-1)\lambda - \cos(m-2)\lambda \end{aligned} \tag{46}$$

These relationships are useful for point calculations but are inefficient for use if a set of points at a uniform longitude interval are being used.

Geodetic Constants

For the evaluation of the reference potential coefficients, the geocentric radius vector etc, we need to adopt a set of constants. We used the values of the Geodetic Reference System 1980. We have:

$$\begin{aligned} a &= 6378137 \text{ meters} \\ kM &= 3986005 \times 10^{8}\ m^{3}\ s^{-2} \\ \omega &= 7292115 \times 10^{-11} \text{ rad } s^{-1} \\ e^{2} &= 0.006\ 694\ 380\ 022\ 90 \\ f &= 0.003\ 352\ 810\ 681\ 18 \\ \gamma_e &= 9.780\ 326\ 7715\ ms^{-2} \\ k' &= 0.001\ 931\ 851\ 353 \end{aligned}$$

These constants are used in the calculation of the reference potential coefficients (for a flattening that is read into the program), the geocentric radius, and normal gravity. These constants can easily be changed in the program.

It is critical to note that the use of the above constants does not mean that the geoid undulation (for example) refers to the GRS80 reference ellipsoid. This is because the zero order term in T has been set to zero. The real reference ellipsoid is that one which best fits the geoid and this may or may not be GRS80.

The Program

The program written to implement the equations previously described is given in Appendix A. This Fortran program was run on an Amdahl 470 V-8 machine using double precision computations.

The program is currently designed for point by point calculation. In this case the input information is as follows:

1. NMAX, F (I3,F10.4)

   NMAX is the highest degree to be used in the expansion, F is 1/f which is the inverse flattening of the reference ellipsoid to which the computed quantities are to be referred.

2. The fully normalized potential coefficients are read from tape or disk file in the form of $(n,m,\bar{C}_{nm},\bar{S}_{nm})$. The arrangement of the input is in order of degree, i.e. from lowest to highest degree. However the storage location for the coefficients is computed from the given n and m values. In this program all coefficiecns are stored in double precision. Space can be saved by storing in single precision.

3. The coordinates of the points at which ζ and the other quantities are to be computed. Specificially (ϕ, λ, h) where ϕ is the geodetic latitude, λ is the longitude and h is the height above the reference ellipsoid. The current format is (3F10.1). The last point is signaled by an end of

file (/*) card. Points having the same latitude should be grouped together as in this case the associated Legendre functions and their derivatives are not re-computed.

The output is printed across the page under column headings: LAT, LON, HEIGHT, UNDU, ANOM, DIST, XI, ETA. Although the output is given two decimal digits, actual accuracy is considerably poorer than this because of the errors in the potential coefficients.

The values computed by this program have been checked against another program written by Tscherning and Goad (1982, private communication). All values checked agreed to two decimal digits.

The computer time needed for a single point calculation (after the potential coefficients are input) is 0.46 seconds with an expansion complete to degree 180 on an Amdahl 470 V/8. A calculation of points on a $12°\times 12°$ grid at 1° intersection took 21.9 seconds. If a limited grid of undulations or anomalies are to be generated the program described by Rizos is the most efficient procedure to date. If a global grid is being generated the program SSYNTH described in Colombo (1981) is the most efficient.

For checking the results of the program the values of ζ , Δg , δ , ξ , η have been computed at five test points using three different sets of potential coefficients to degree 180. These values have been computed with respect to an ellipsoid which has the flattening of GRS80 and are given in Table 1.

Table 1

Sample Computed Values

(reference flattening = 1/298.257222)

$\phi°$	$\lambda°$	h(m)		ζ(m)	Δg(mgals)	δ(mgals)	ξ''	η''
21°	1°	0	Rapp78	34.46	20.07	30.65	0.68	-0.25
			Rapp81	30.56	7.73	17.11	0.60	0.40
			GEM10C	28.37	4.12	12.83	-0.10	0.21
21°	45°	0	Rapp78	-11.19	-4.75	-8.19	-5.41	11.12
			Rapp81	-9.58	-5.55	-8.49	-4.24	10.63
			GEM10C	-9.68	-8.46	-11.43	-2.23	8.98
5°	79°	0	Rapp78	-104.42	-84.60	-116.62	-1.43	0.64
			Rapp81	-107.48	-91.84	-124.81	0.02	0.65
			GEM10C	-106.20	-87.66	-120.23	-1.13	-1.04
5°	79°	10000	Rapp78	-103.58	-78.90	-110.52	-1.63	0.35
			Rapp81	-106.58	-85.49	-118.02	-0.22	0.50
			GEM10C	-105.35	-80.51	-112.66	-1.12	-0.93
87°	21°	0	Rapp78	15.43	-1.46	3.32	1.32	2.37
			Rapp81	20.23	8.86	15.12	0.81	1.86
			GEM10C	18.38	3.58	9.26	2.59	4.05

Summary

This report describes a Fortran computer program that can be used for the calculation of ζ , Δg , δ , ξ , η which are dependent on a set of fully normalized potential coefficients. The program has been set to work to degree 180 and it can be extended higher.

The equations used for the calculations are to some extent spherical approximations. However literal interpretation of certain quantities would be formally correct (e.g. the radial component of the gravity disturbance). Correction terms for spherical harmonic expansions evaluated considering the ellipticity of the earth are described to some extent, in Jekeli (1981, section 4).

The input quantities to the program are geodetic latitude, longitude and height above the ellipsoid. In theory these quantities should be given with respect to a geocentric ellipsoid. In practice the use of non-geocentric coordinates would cause small but systematic errors in the results.

The computed quantities refer to a geocentric ellipsoid whose flattening is an input parameter. The size of this ellipsoid is not specifically defined because the zero degree term in the disturbing potential expansion has been set to zero. In most applications the equatorial radius of the ellipsoid is the current best estimate.

The computer program of this report has been checked against other programs with excellent agreement. The stability of the algorithms for the associated Legendre functions has been checked by Colombo (1981) and by Singh (1982). For some applications at high latitude underflows may occur in the computations. These are machine dependent quantities and can be turned off if desired.

Other procedures have been developed that extend the derivatives of the potential to the second derivative (Tscherning and Poder, 1981). In addition, problems at the pole that exist with our current program (for the derivative of $\bar{P}_{nm}$) are avoided with the Tscherning/Poder application of the Clenshaw summation.

References

Colombo, O., Numerical Methods for Harmonic Analysis on the Sphere, Report No. 310, Dept. of Geodetic Science and Surveying, The Ohio State University, Columbus, 1982.

Cook, A.H., The External Gravity Field of a Rotating Spheroid to the order of e^3, Geophysical Journal, Vol.2, No.3, 199-214, 1959.

Heiskanen, W., and H. Moritz, Physical Geodesy, W. Freeman, San Francisco, 1967.

Jekeli, D., The Downward Continuation to the Earth's Surface of Truncated Spherical and Ellipsoidal Harmonic Series of the Gravity and Height Anomalies, Report No. 323, Dept. of Geodetic Science and Surveying, The Ohio State University, Columbus, 144p, December 1981.

Lerch, F.J., B. Putney, C. Wagner and S. Klosko, Goddard Earth Models for Oceanographic Applications, Marine Geodesy, Vol. 5, No. 2, 1981.

Pick, M., J. Picha, V. Vyskočil, Theory of the Earth's Gravity Field, Elsevier Scientific Publishing Co., Amsterdam, 1973.

Rapp, R.H., A Global 1°x 1° Anomaly Field Combining Satellite Geos-3 Altimmeter and Terrestrial Anomaly Data, Dept. of Geodetic Science Report No. 278, 1978.

Rapp, R.H., Geometric Geodesy, Volume I (Basic Principles), The Ohio State University, Columbus, 1981

Rapp, R.H., The Earth's Gravity Field to Degree and Order 180 Using Seasat Altimeter Data, Terrestrial Gravity Data, and Other Data, Dept. of Geodetic Science and Surveying, The Ohio State University, Columbus, Report No. 322, December 1981.

Rizos, C., An Efficient Computer Technique for the Evaluation of Geopotential from Spherical Harmonic Models, Australian Journal of Geodesy, Photogrammetry and Surveying, 161-170, No. 31, 1979.

Singh, A., On Numerical Evaluation of Normalized Associated Legendre Functions, internal report, Dept. of Geodetic Science and Surveying, The Ohio State University, 1982.

Tscherning, C.C. and K. Poder, Some Geodetic Applications of Clenshaw summation, paper presented at VIII Symposium on Mathematical Geodesy, Como, Italy, Sept. 1981.

Appendix

The FORTRAN Program

```
//   JOB  'XXXXX,XXXXXXXXX','RAPP,R.H.',
//   TIME=(0,40),REGION=1024K
/*JOBPARM  LINES=3900,DISKIO=2400,V=R
//S1   EXEC FORTQCG
//FORT.SYSIN DD *
C THIS PROGRAM WAS PUT IN ITS PRESENT FORM BY R.H. RAPP IN AUG 1982
C THE PROGRAM IS A MODIFICATION OF PROGRAM F379
C    POINT COMPUTATION FROM HARMONIC COEFFICIENTS
C    DIMENSIONS OF P,Q,HC,HS MUST BE AT LEAST ((MAXN+1)*(MAXN+2))/2,
C    DIMENSIONS OF SINML,COSML,SCRAP MUST BE AT LEAST MAXN,
C          WHERE MAXN IS MAXIMUM ORDER OF COMPUTATION
C THE CURRENT DIMENSIONS ARE SET FOR A MAXIMUM DEGREE OF 180
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*8  P(16471),SCRAP(181),RLEG(181),DLEG(181),RLNN(181),
     *PDER(16471),SINML(181),COSML(181)
      REAL*8 HC(16471),HS(16471)
      DATA RAD/57.295779513082320O/
C    F IS THE REFERENCE INVERSE FLATTENING,NMAX IS THE MAXIMUM DEGREE
010   READ(5,900) NMAX,F
900   FORMAT(I3,F10.4)
      WRITE(6,910) NMAX,F
910   FORMAT(1H1,//' MAXIMUM DEGREE =',I4,30X,'1/F=',F8.4///)
      F=1.0D0/F
      CALL DHCSIN(NMAX,F,RJ2,RJ4,RJ6,HC,HS)
C SETTING IFLAG=0 FORCES LEGENDRE FUNCTION DERIVATIVES TO BE TAKEN
      IFLAG=0
      IR=0
      K=NMAX+1
      FLATL=90.0D0
      WRITE(6,978)
 978  FORMAT(1H1,//'    LAT   ','    LON   ','  HEIGHT ','      UNDU   ',
     *'    ANOM    ','    DIST   ','     XI    ','    ETA')
C READ GEODETIC LATITUDE,LONGITUDE,AND HEIGHT ABOVE THE ELLIPSOID
 030  READ(5,930,END=090) FLAT,FLON,HT
 930  FORMAT(3F10.1)
C COMPUTE THE GEOCENTRIC LATITUDE,GEOCENTRIC RADIUS,NORMAL GRAVITY
      CALL RADGRA(FLAT,FLON,HT,RLAT,GR,RE)
      IF(FLATL.EQ.FLAT) GO TO 040
      RLAT1=RLAT
      RLAT=1.5707963267948966D0-RLAT
      FLATL=FLAT
      DO 25 J=1,K
      M=J-1
      CALL LEGFDN(M,RLAT,RLEG,DLEG,NMAX,IR,RLNN,IFLAG)
      DO 26 I=J,K
      N=I-1
      LOC=(N*(N+1))/2+M+1
      PDER(LOC)=DLEG(I)
  26  P(LOC)=RLEG(I)
  25  CONTINUE
040   RLON=FLON/RAD
      CALLDSCML (RLON,NMAX,SINML,COSML)
      CALL HUNDU(U,DG,DIS,XI,ETA,HT,NMAX,P,PDER,HC,HS,SINML,COSML,
     *GR,RE,RLAT1)
      WRITE(6,940) FLAT,FLON,HT,U,DG,DIS,XI,ETA
 940  FORMAT(2F9.4,1F09.2,5F10.2)
      GO TO 30
 90   STOP
      END
      SUBROUTINE HUNDU(UNDU,ANOM,DIST,XI,ETA,HT,NMAX,P,PDER,HC,HS,
     *SINML,COSML,GR,RE,ANG)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION SINML(1),COSML(1),P(1),PDER(1)
      REAL*8 HC(1),HS(1)
```

```
C   CONSTANTS FOR GRS80
       DATA GM/.3986005D15/,AE/6378137.0D0/,RHO/206264.806D0/
       AR=AE/RE
       ARN=AR
       A=0.0
       B=0.0
       XI=0.0
       ETA=0.0
       K=3
       DO 030 N=2,NMAX
       ARN=ARN*AR
       K=K+1
       SUM=P(K)*HC(K)
       SUM1=PDER(K)*HC(K)
       SUM2=0.0
       DO 020 M=1,N
       K=K+1
       TEMP=HC(K)*COSML(M)+HS(K)*SINML(M)
       SUM1=SUM1+PDER(K)*TEMP
       SUM2=SUM2+P(K)*M*(-HC(K)*SINML(M)+HS(K)*COSML(M))
020    SUM=SUM+      P(K)  *TEMP
       B=B+SUM*ARN*(N-1)
       XI=XI+SUM1*ARN
       ETA=ETA+SUM2*ARN
 30    A=A+SUM*ARN
       UNDU=A*GM/(GR*RE)
       ANOM=B*GM/RE**2*1.D5
       DIST=ANOM+2.*UNDU*GR/RE*1.D5
C THE SIGN OF + IN XI OCCURS DUE TO THE DERIVATIVE BEING
C WITH RESPECT TO POLAR DISTANCE NOT LATITUDE
       XI=+RHO*GM/(GR*RE**2)*XI
       ETA=-RHO*GM/(GR*RE**2*DCOS(ANG))*ETA
C THE UNITS OF THE UNDULATION ARE METERS
C THE UNITS OF THE ANOMALY AND DISTURBANCE ARE MGALS
C THE UNITS OF THE DEFLECTIONS ARE SECONDS
       RETURN
       END
       SUBROUTINE  DSCML  (RLON,NMAX,SINML,COSML)
       IMPLICIT REAL*8 (A-H,O-Z)
       DIMENSION SINML(1),COSML(1)
       A=DSIN(RLON)
       B=DCOS(RLON)
       SINML(1)=A
       COSML(1)=B
       SINML(2)=2.0*B*A
       COSML(2)=2.0*B*B-1.0
       DO 010 M=3,NMAX
       SINML(M)=2.0*B*SINML(M-1)-SINML(M-2)
010    COSML(M)=2.0*B*COSML(M-1)-COSML(M-2)
       RETURN
       END
       SUBROUTINE DHCSIN (NMAX,F,J2,J4,J6,HC,HS)
       IMPLICIT REAL*8 (A-H,O-Z)
       REAL*8 J2,J4,J6
       REAL*8 HC(1),HS(1)
C        THIS VERSION USES IMPROVED J2,J4,J6 FROM COOK PAPER(1959)
C   CONSTANTS FROM GRS 80
       DATA FKM,OM,A/3.986005D14,7.292115D-5,6378137.0D0/
       FM=OM**2*A**3*(1.0D0-F)/FKM
       M=((NMAX+1)*(NMAX+2))/2
       DO 001 N=1,M
       HC(N)=0.0
001    HS(N)=0.0
 02    READ(12,END=3)N,M,C,S
       IF(N.GT.NMAX) GO TO 003
```

```
      N=(N*(N+1))/2+M+1
      HC(N)=C
      HS(N)=S
      GO TO 002
    3 J2=2.0D0/3.0D0*(F*(1.0D0-F/2.0D0)-FM/2.0D0*(1.0D0-2.0D0/7.0D0*F
     *+11.0D0*F*F/49.0D0))
      J4=-4.0D0/35.0D0*F*(1.0D0-F/2.0D0)*(7.0D0*F*(1.0D0-F/2.0D0)
     *-5.0D0*FM*(1.0D0-2.0D0*F/7.0D0))
      J6=4.*F**2*(6.*F-5.*FM)/21.
      HC(4)=HC(4)+J2/DSQRT(5.D0)
      HC(11)=HC(11)+J4/3.0D0
      HC(22)=HC(22)+J6/DSQRT(13.D0)
      RETURN
      END
      SUBROUTINE LEGFDN(M,THETA,RLEG,DLEG,NMX,IR,RLNN,IFLAG)
C
C             THIS SUBROUTINE COMPUTES  ALL NORMALIZED LEGENDRE FUNCTIONS
C             IN "RLEG" AND THEIR DERIVATIVES IN "DLEG". ORDER IS ALWAYS
C             M , AND COLATITUDE IS ALWAYS THETA  (RADIANS). MAXIMUM DEGR
C             IS  NMX  . ALL CALCULATIONS IN DOUBLE PRECISION.
C             IR  MUST BE SET TO ZERO BEFORE THE FIRST CALL TO THIS SUB.
C             THE DIMENSIONS OF ARRAYS  RLEG, DLEG, AND RLNN  MUST BE
C             AT LEAST EQUAL TO  NMX+1  .
C
C             THIS PROGRAM DOES NOT COMPUTE DERIVATIVES AT THE POLES .
C
C             IF    IFLAG = 1  , ONLY THE LEGENDRE FUNCTIONS ARE
C             COMPUTED.
C
C      ORIGINAL PROGRAMMER :OSCAR L. COLOMBO, DEPT. OF GEODETIC SCIENCE,
C      THE OHIO STATE UNIVERSITY, AUGUST 1980 . ******************
C
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION RLEG(1),DLEG(1),RLNN(1)
     2, DRTS(1300),DIRT(1300)
      NMX1 = NMX+1
      NMX2P = 2*NMX+1
      M1 = M+1
      M2 = M+2
      M3 = M+3
      IF(IR.EQ.1) GO TO 10
      IR = 1
      DO 5      N = 1,NMX2P
      DRTS(N) = DSQRT(N*1.D0)
    5 DIRT(N) = 1.D0/DRTS(N)
   10 COTHET = DCOS(THETA)
      SITHET = DSIN(THETA)
      IF(IFLAG.NE.1.AND.THETA.NE.0.D0)SITHI = 1.D0/SITHET
C
C             COMPUTE THE LEGENDRE FUNCTIONS .
C
      RLNN(1) = 1.D0
      RLNN(2) = SITHET*DRTS(3)
      DO 15     N1 = 3,M1
      N = N1-1
      N2 = 2*N
   15 RLNN(N1) = DRTS(N2+1)*DIRT(N2)*SITHET*RLNN(N1-1)
      IF(M.GT.1) GO TO 20
      IF(M.EQ.0) GO TO 16
      RLEG(2) = RLNN(2)
      RLEG(3) = DRTS(5)*COTHET*RLEG(2)
      GO TO 20
   16 RLEG(1) = 1.D0
      RLEG(2) = COTHET*DRTS(3)
```

```
   20 CONTINUE
      RLEG(M1) = RLNN(M1)
      RLEG(M2) = DRTS(M1*2+1)*COTHET*RLEG(M1)
      DO 30     N1 = M3,NMX1
      N = N1-1
      IF(M.EQ.0.AND.N.LT.2.OR.M.EQ.1.AND.N.LT.3) GO TO 30
      N2 = 2*N
      RLEG(N1) = DRTS(N2+1)*DIRT(N+M)*DIRT(N-M)*(DRTS(N2-1)*COTHET*
     2 RLEG(N1-1)-DRTS(N+M-1)*DRTS(N-M-1)*DIRT(N2-3)*RLEG(N1-2))
      GO TO 30
   30 CONTINUE
      IF(IFLAG.EQ.1) RETURN
      IF(SITHET.EQ.0.D0) WRITE(6,99)
   99 FORMAT(//' *** LEGFDN  DOES NOT COMPUTE DERIVATIVES AT THE POLES
     2 *****************'//)
      IF(SITHET.EQ.0.D0) RETURN
C
C               COMPUTE ALL THE DERIVATIVES OF THE LEGENDRE FUNCTIONS.
C
      RLNN(1) = 0.D0
      RLN = RLNN(2)
      RLNN(2) = DRTS(3)*COTHET
      DO 40        N1 = 3, M1
      N = N1-1
      N2 = 2*N
      RLN1 = RLNN(N1)
      RLNN(N1) = DRTS(N2+1)*DIRT(N2)*(SITHET*RLNN(N)+COTHET*RLN)
      RLN = RLN1
   40 CONTINUE
      DLEG(M1) = RLNN(M1)
      DO 60        N1 = M2,NMX1
      N = N1-1
      N2 = N*2
      DLEG(N1) = SITHI*( N    *RLEG(N1)*COTHET-DRTS(N-M)*DRTS(N+M)*
     2 DRTS(N2+1)*DIRT(N2-1)*RLEG(N))
   60 CONTINUE
      RETURN
      END
      SUBROUTINE RADGRA(FLAT,FLON,HT,RLAT,GR,RE)
      IMPLICIT REAL*8(A-H,O-Z)
C THIS SUBROUTINE COMPUTES GEOCENTRIC DISTANCE TO THE POINT,
C THE GEOCENTRIC LATITUDE,AND
C AN APPROXIMATE VALUE OF NORMAL GRAVITY AT THE POINT BASED
C ON CONSTANTS OF THE GEODETIC REFERENCE SYSTEM 1980
      DATA AE/6378137.D0/,E2/.00669438002290D0/,RAD/57.29577951308232D0/
      REAL*8 N
      FLATR=FLAT/RAD
      FLONR=FLON/RAD
      T1=DSIN(FLATR)**2
      N=AE/DSQRT(1.-E2*T1)
      T2=(N+HT)*DCOS(FLATR)
      X=T2*DCOS(FLONR)
      Y=T2*DSIN(FLONR)
      Z=(N*(1.-E2)+HT)*DSIN(FLATR)
      N=AE/DSQRT(1.-E2*T1)
C COMPUTE THE GEOCENTRIC RADIUS
      RE=DSQRT(X**2+Y**2+Z**2)
C COMPUTE THE GEOCENTRIC LATITUDE
      RLAT=DATAN(Z/DSQRT(X**2+Y**2))
C COMPUTE NORMAL GRAVITY:UNITS ARE M/SEC**2
      GR=9.7803267715D0*(1.+.001931851353D0*T1)/DSQRT(1.-E2*T1)
C CORRECT FOR ELEVATION IN AN APPROXIMATE WAY
      GR=GR-HT*0.3086D-5
      RETURN
      END
```

```
//*
//GO.FT12F001 DD UNIT=USERDA,DISP=(OLD,KEEP),
// DSN=TS0453.D1978.POT.COEF.TO180
//GO.SYSIN DD *
180298.25722
21.0        1.0
21.0        45.0
5.0         79.0
5.0         79.0        10000.0
87.0        21.0
/*
```

Sample Output

LAT	LON	HEIGHT	UNDU	ANOM	DIST	XI	ETA
21.0000	1.0000	0.0	34.46	20.07	30.65	0.68	-0.25
21.0000	45.0000	0.0	-9.76	-2.82	-5.81	-5.25	11.63
5.0000	79.0000	0.0	-104.42	-84.60	-116.62	-1.43	0.64
5.0000	79.0000	10000.00	-103.58	-78.90	-110.52	-1.63	0.35
87.0000	21.0000	0.0	15.43	-1.46	3.32	1.32	2.37